# Maintainer: Antonio Rojas <arojas@archlinux.org>

pkgname=python-pyqt5-sip
pkgver=12.13.0
pkgrel=1
arch=(loong64 x86_64)
pkgdesc='The sip module support for PyQt5'
url='https://riverbankcomputing.com/software/pyqt/intro'
license=(GPL)
depends=(glibc
         python)
makedepends=(python-build
             python-installer
             python-setuptools
             python-wheel)
source=(https://pypi.python.org/packages/source/P/PyQt5-sip/PyQt5_sip-$pkgver.tar.gz)
sha256sums=('7f321daf84b9c9dbca61b80e1ef37bdaffc0e93312edae2cd7da25b953971d91')

build() {
  cd PyQt5_sip-$pkgver
  python -m build --wheel --no-isolation
}

package()  {
  cd PyQt5_sip-$pkgver
  python -m installer --destdir="$pkgdir" dist/*.whl
}
